
Writing PromEx Plugins
    

This guide will walk you through writing a PromEx plugin. Whether this plugin is for a dependent library or for your
internal application metrics, the same patterns apply.

  
    
  
  Getting started


In order for PromEx to be able to load the appropriate metrics from your plugins, your modules need to leverage the
PromEx behaviour. This behaviour defines 3 optional callbacks. Those callbacks are:
	event_metrics/1
	polling_metrics/1
	manual_metrics/1

Each of these callbacks is supposed to return a list of metrics of that type. For example, polling_metrics/1 needs to
return a list of PromEx.MetricTypes.Polling structs (a single struct is also an acceptable return). By doing this, your
plugin can load your metrics and deal with the nuances of each metric type properly. Each of the MetricTypes structs
all have a field called :metrics. This field contains a list of all the Telemetry.Metrics definitions that were
provided to the struct build function.

  
    
  
  Adding Event Metrics


To have your custom plugin expose event based metrics, implement a event_metrics/1 function and build out a collection
of Telemetry.Metrics structs (distribution, counter, last_value, and sum). Be sure to look at plugins like
PromEx.Plugins.Phoenix for more in depth examples.
defmodule MyApp.PromEx.Plugins.MyPhoenix do
  use PromEx.Plugin

  @impl true
  def event_metrics(opts) do
    http_metrics_tags = gen_http_metrics_tags(opts)
    phoenix_router = get_phoenix_router(opts)
    phoenix_stop_event = [:phoenix, :endpoint, :stop]

    Event.build(
      :phoenix_http_event_metrics,
      [
        # Capture request duration information
        distribution(
          [:phoenix, :http, :request, :duration, :milliseconds],
          event_name: phoenix_stop_event,
          measurement: :duration,
          description: "The time it takes for the application to respond to HTTP requests.",
          reporter_options: [
            buckets: exponential!(1, 2, 12)
          ],
          tag_values: get_conn_tags(phoenix_router),
          tags: http_metrics_tags,
          unit: {:native, :millisecond}
        )

        # Additional event based metrics ...
      ]
    )
  end
end

  
    
  
  Adding Polling Metrics


Polling metrics are similar to event metrics in that they require similar fields (group_name and metrics to be
specific). In addition, the PromEx.MetricTypes.Polling.build/4 function requires an measurements_mfa argument which
specifies what function will be executed on the polling interval. This function should run :telemetry.execute/3
somewhere in its function body. Once that event is executed, the corresponding event in the struct will be triggered and
you will capture the desired data point. The following example from PromEx.Plugins.Beam should highlight this concept:
defmodule PromEx.Plugins.Beam do
  use PromEx.Plugin

  @memory_event [:prom_ex, :plugin, :beam, :memory]

  @impl true
  def polling_metrics(opts) do
    poll_rate = Keyword.get(opts, :poll_rate, 5_000)

    [
      memory_metrics(poll_rate)
    ]
  end

  defp memory_metrics(poll_rate) do
    Polling.build(
      :beam_memory_polling_events,
      poll_rate,
      {__MODULE__, :execute_memory_metrics, []},
      [
        # Capture the total memory allocated to the entire Erlang VM (or BEAM for short)
        last_value(
          [:beam, :memory, :total, :kilobytes],
          event_name: @memory_event,
          description: "The total amount of memory currently allocated.",
          measurement: :total,
          unit: {:byte, :kilobyte}
        )

        # More memory metrics here
      ]
    )
  end

  @doc false
  def execute_memory_metrics do
    memory_measurements =
      :erlang.memory()
      |> Map.new()

    :telemetry.execute(@memory_event, memory_measurements, %{})
  end
end
Depending on what :poll_rate value you pass to the initialization tuple for PromEx.Plugins.Beam, the
execute_memory_metrics/0 function will be execute on that specified interval.

  
    
  
  Adding Manual Metrics


Manual metrics behave more or less the same as polling metrics except they do not require a poll rate value. Instead the
provided measurements_mfa is called once on application start, and the metrics are only then updated if you make a
call to PromEx.ManualMetricsManager.refresh_metrics/1. An example of this can be seen from the
PromEx.Plugins.Application plugin:
defmodule PromEx.Plugins.Application do
  use PromEx.Plugin

  @impl true
  def manual_metrics(opts) do
    otp_app = Keyword.fetch!(opts, :otp_app)
    apps = Keyword.get(opts, :deps, :all)

    Manual.build(
      :application_versions_manual_metrics,
      {__MODULE__, :apps_running, [otp_app, apps]},
      [
        # Capture information regarding the primary application (i.e the user's application)
        last_value(
          [otp_app | [:application, :primary, :info]],
          event_name: [otp_app | [:application, :primary, :info]],
          description: "Information regarding the primary application.",
          measurement: :status,
          tags: [:name, :version, :modules]
        )

        # Additional metrics here
      ]
    )
  end

  @doc false
  def apps_running(otp_app, apps) do
    ...

    # Emit primary app details
    :telemetry.execute(
      [otp_app | [:application, :primary, :info]],
      %{
        status: if(Map.has_key?(started_apps, otp_app), do: 1, else: 0)
      },
      %{
        name: otp_app,
        version:
          Map.get_lazy(started_apps, otp_app, fn ->
            Map.get(loaded_only_apps, otp_app, "undefined")
          end),
        modules: length(Application.spec(otp_app)[:modules])
      }
    )
  end
end
So in this example, apps_running/2 is the function that is denoted by the MFA and will be called once automatically on
application start, but then at that point it is up to the user to refresh the data point.

Introduction to Telemetry
    

This section serves as a quick primer for the Telemetry library, how it works, and how you can leverage it in your
applications. While there is nothing specific to PromEx here, but it is important to know how Telemetry works so that
you can effectively create your own PromEx plugins.

  
    
  
  What is Telemetry?


At a high level, Telemetry offers a means for libraries and applications to surface internal events. These events can
be emitted prior to starting some logical body of work, after the work has completed, after the processing of the work
has resulted in an exception, or even any time you want to denote that something of importance has occurred. As an
example, these events can include the time it takes to process incoming HTTP requests (like in Phoenix) or the time
it takes to process a job (like in Broadway & Oban) and well as metadata related to the event (like the requested route
in Phoenix).

  
    
  
  How Does it Work?


Under the hood, Telemetry works by attaching callback functions to events (tracked in ETS), and then invoking those
functions serially whenever that event occurs. In other words, any time an event is triggered, Telemetry dynamically
dispatches to each of the callbacks that are registered for that particular event. You can think of this as effectively
being a pub/sub style library but synchronous as opposed to asynchronous. Given that the callbacks are executed in a
synchronous fashion, it is highly recommended that you keep your callbacks very lightweight and functionally limited. If
additional blocking work needs to take place as a result of an event, be sure to spin that work off into its own Task or
pass it along to a separate GenServer.

  
    
  
  Why is Telemetry Important?


In my opinion, Telemetry (and the other repositories in the beam-telemetry GitHub organization) provides two key benefits
to the Erlang, Elixir and Beam communities.
Firstly, Telemetry provides a consistent interface through which applications and libraries can expose internal events.
This consistent interface consists of measurements and metadata that are attached to each event that can then be used by
consumers in whichever way fits the user's needs. For example, using the same Telemetry event, you can produce a custom
log message and even metrics by attaching two separate callback functions to the desired Telemetry event.
Secondly, attaching callbacks to Telemetry events is a very low friction operation and can be done without much
ceremony. This is important given that if the ergonomics of Telemetry were cumbersome to work with, library authors
would not be inclined to leverage Telemetry for surfacing internal events. Specifically, Telemetry does not require
any global setup or configuration by either the library author or the user. Instead, Telemetry will sort out the dynamic
function dispatch through ETS tables that it manages and will only invoke callback functions when callbacks have been
attached to a particular event.
As a result of these two points (and others), it is no surprise that over 100 libraries in the Elixir and Erlang
ecosystem have adopted Telemetry as their primary means of surfacing internal events.

  
    
  
  How Can I Use Telemetry in my Project?


As previously mentioned, leveraging Telemetry consists of a library (or your application) executing an event, and you as
the user attaching a callback to said event. As an example, let's suppose that we want to emit an event any time a user
registers for our service. In our registration function we could do something like so in order to capture when users are
successfully created and when errors are encountered:
def register_user(attrs) do
  %User{}
  |> User.registration_changeset(updated_attrs)
  |> Repo.insert()
  |> case do
    {:ok, new_user} = result ->
      :telemetry.execute([:my_cool_app, :accounts, :new_user, :success], %{}, %{user: new_user})
      result

    {:error, changeset} = error ->
      :telemetry.execute([:my_cool_app, :accounts, :new_user, :error], %{}, %{error: changeset})
      error
  end
end
Elsewhere in your application code (perhaps where you initialize your application) you can attach to these events by
doing the following:
:telemetry.attach(
  "my-handler-1",
  [:my_cool_app, :accounts, :new_user, :success],
  fn _event_name, _event_measurement, event_metadata, _config ->
    Logger.debug("User has registered: #{inspect(event_metadata)}")
  end,
  %{}
)

:telemetry.attach(
  "my-handler-2",
  [:my_cool_app, :accounts, :new_user, :error],
  fn _event_name, _event_measurement, event_metadata, _config ->
    Logger.warning("User failed to register: #{inspect(event_metadata)}")
  end,
  %{}
)
Now, any time an error occurs and a user fails to register a warning log is generated with the metadata related to the
error. Similarly, a debug log message is created any time a user is successfully created.

  
    
  
  Additional Resources


	https://github.com/beam-telemetry/telemetry
	https://hexdocs.pm/telemetry_metrics/Telemetry.Metrics.html
	https://keathley.io/blog/telemetry-conventions.html



  

  
    
    Running Multiple Agents - PromEx v1.10.0
    
    

    


  
  

    
Running Multiple Agents
    

Suppose you want to send your metrics to two separate Prometheus databases, and define two separate sets of dashboards - one for the ops team, and one for the business team.
You can do this by creating two PromEx configurations.
To generate distinct PromEx modules:
mix prom_ex.gen.config -d ops -m PromExOps
mix prom_ex.gen.config -d biz -m PromExBiz

Each PromEx config will run its own Grafana Agent. You need to configure working_directory, agent_port and grpc_port to make sure they don't collide:
config :my_app, MyApp.PromExOps,
  grafana_agent: [
    working_directory: System.fetch_env!("RELEASE_TMP") <> "/grafana-ops",
    config_opts: [
      ...
      agent_port: 4040,
      grpc_port: 9040
    ]
  ]

config :my_app, MyApp.PromExBiz,
  grafana_agent: [
    working_directory: System.fetch_env!("RELEASE_TMP") <> "/grafana-biz",
    config_opts: [
      ...
      agent_port: 4041,
      grpc_port: 9041
    ]
  ]
Add each module to your application's supervisor per the directions in the generated PromEx file.

  
    
  
  Endpoints


You can configure each agent to scrape the same set of metrics:
# endpoint.ex
plug PromEx.Plug, prom_ex_module: MyApp.PromExOps

# in mix config, set `metrics_server_path: "/metrics"`
or define separate endpoints:
plug PromEx.Plug, prom_ex_module: MyApp.PromExOps, path: "/metrics/ops"
plug PromEx.Plug, prom_ex_module: MyApp.PromExBiz, path: "/metrics/biz"

Seeding Metrics
    

In some applications, telemetry is only emitted periodically or in spurts. Capturing
these metrics via PromEx.Plugin.event_metrics/1 can lead to gaps in Grafana graphs
after a service restart or deployment.
When an application starts, its Prometheus metrics are empty; its metrics_server_path
will only show metrics that have been emitted and captured by PromEx over the course
of the beam's lifespan. When a metric is not seen within a time period, Prometheus
returns no data for that metric, and in Grafana rather than data points with 0 values,
the graph will show No Data.
One mechanism to ensure that the graphs are populated is to seed the event metrics, i.e.
emit them at least once after PromEx has started. This can be done via a
Application.start_phase/3.

  
    
  
  Start Phases


Start phases are configured in a project's application callback in mix.exs:
  def application do
    [
      extra_applications: [],
      mod: {MyApp.Application, []},
      start_phases: [seed_prom_ex_telemetry: []]
    ]
  end
This requires adding a callback to MyApp.Application:
defmodule MyApp.Application do
  use Application

  @impl true
  def start(_type, _args) do
    children = [
      MyApp.PromEx
      # ... Repo, Phoenix, etc
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  @impl true
  def start_phase(:seed_prom_ex_telemetry, :normal, _) do
    MyApp.PromEx.seed_event_metrics()
    :ok
  end
end
The seed_event_metrics/0 function can send any telemetry events desired, with any
variations in measurements and metadata to ensure that the metrics_server_path is
fully populated when it is first scraped.
def seed_event_metrics do
  # :telemetry.execute([:my_app, :event], %{}, %{status: :ok})
  # :telemetry.execute([:my_app, :event], %{}, %{status: :error})
  # :telemetry.execute([:my_app, :event_span, :start], %{count: 1}, %{})
  # :telemetry.execute([:my_app, :event_span, :stop], %{count: 1}, %{})
  # :telemetry.execute([:my_app, :event_span, :exception], %{count: 1}, %{})

  :ok
end
One thing to note is that for counter metrics, seeding the events in this way will
result in the values being initialized to a count of 1. In Grafana graphs showing
counters over time, the value resetting to either 0 or 1 will result in a negative
delta; one may want to combine idelta with clamp_min
(i.e. clamp_min(idelta(<event_name>[$__rate_interval]), 0)) to ensure that the
initial value after a restart appears in the graph as no change.

  Application


The application dashboard surfaces static information regarding the application. Things such as the dependencies of the
application, GIT SHA+author and uptime.
[image: Application Dashboard]
  
    
  
  BEAM


The BEAM dashboard presents information regarding the Erlang virtual machine. Things such as memory statistics,
process and atom counts, scheduler information, and much more
[image: BEAM Dashboard]
  
    
  
  Ecto


The Ecto dashboard contains charts to track query execution time, number of results returned, and static information
regarding the configured Repo.
[image: Ecto Dashboard]
  
    
  
  Oban


The Oban plugin presents information regarding the job execution time, job queue time, available jobs in queue, static
configuration settings and much, much more.
[image: Oban Dashboard]
  
    
  
  Phoenix


The Phoenix dashboard presents information regarding HTTP requests and channel connections.
[image: Phoenix Dashboard]
  
    
  
  Phoenix LiveView


The Phoenix LiveView dashboard presents information regarding the various LiveView callbacks and the time it takes to
execute them.
[image: Phoenix LiveView Dashboard]
  
    
  
  Broadway


The Broadway dashboard presents information regarding the messages the Broadway has processed as well as how
it has processed message batches.
[image: Broadway Dashboard]

PromEx is a plugin based library which can be used to capture
telemetry events and report them out for consumption by Prometheus.
The main purpose of this particular library is to provide the
behaviour that all PromEx plugins leverage so that a consistent
interface can be achieved and so that leveraging multiple plugins is
effortless from the user's point of view.
To use PromEx you need to define a module that uses the PromEx library. This module
will also need to have some application config set for it similarly to how Ecto does.
For example, for a PromEx module defined like so:
defmodule MyApp.PromEx do
  use PromEx,  otp_app: :web_app

  ...
end
You would have an application configuration set like so:
config :my_app, MyApp.PromEx,
  manual_metrics_start_delay: :no_delay,
  drop_metrics_groups: [],
  grafana: [
    host: System.get_env("GRAFANA_HOST", "http://grafana:3000"),
    auth_token: System.get_env("GRAFANA_TOKEN", ""),
    upload_dashboards_on_start: true,
    folder_name: "My App Dashboards",
    annotate_app_lifecycle: true
  ]
The options that you can pass to PromEx macro are outlined in the following section. In order
to tell PromEx what plugins you would like to use and what dashboards you would like PromEx
to upload for you, implement the plugins/0 and dashboards/0 callbacks respectively. The
dashboard_assigns/0 callback will be used when your EEx template Grafana dashboards are
rendered so that the dashboards that are created for your application coincide with the PromEx
configuration for the application. If your dashboards are not EEx templates, then the dashboard
assigns are not passed through. Each plugin also has an accompanying Grafana dashboard that you
can leverage to plot all of the plugin captured data.
In order to expose captured metrics, you can leverage the PromEx provided Plug PromEx.Plug.
See the PromEx.Plug documentation modules for specifics on how to use it.

  
    
  
  Options


	:otp_app - This is a REQUIRED field and is used by PromEx to fetch the application
configuration values for the various PromEx capture modules. Make sure that this value
matches the :app value in project/0 from your mix.exs file. If you use the PromEx
mix prom_ex.create mix task this will be done automatically for you.


  
    
  
  Top level PromEx Configuration


PromEx is generally configured per application per PromEx module as you saw in the previous section. The only
setting that is currently available globally for all instances of PromEx is that of the storage adapter. The
storage adapter configuration determines how PromEx captures and stores metrics. There are currently two
available adapters:
	TelemetryMetricsPrometheus.Core - This is the
default adapter and is included with PromEx.

	Peep - In order to use Peep as your storage adapter,  you will need to add the following
to your config.exs file:
 config :prom_ex, :storage_adapter, PromEx.Storage.Peep



  
    
  
  PromEx Plugins


All metrics collection will be delegated to plugins which can be found here:
Foundational metrics:
	[X] PromEx.Plugins.Application Application related informational metrics
	[X] PromEx.Plugins.Beam BEAM virtual machine metrics
	[ ] Operating System (http://erlang.org/doc/man/os_mon_app.html)

Library metrics:
	[X] PromEx.Plugins.Ecto - Telemetry docs
	[X] PromEx.Plugins.Oban - Telemetry docs
	[X] PromEx.Plugins.Phoenix - Telemetry docs
	[X] PromEx.Plugins.PhoenixLiveView - Telemetry docs
	[X] PromEx.Plugins.Absinthe - Telemetry docs
	[X] PromEx.Plugins.PlugCowboy - Telemetry docs
	[X] PromEx.Plugins.PlugRouter - Telemetry docs
	[X] PromEx.Plugins.Broadway - Telemetry docs

Backlog Elixir library metrics:
	[ ] Finch - Telemetry docs
	[ ] Swoosh - Telemetry docs
	[ ] ChromicPDF - Telemetry docs
	[ ] Dataloader - Telemetry docs
	[ ] GenRMQ - Telemetry docs
	[ ] Plug - Telemetry docs
	[ ] Redix - Telemetry docs
	[ ] Tesla - Telemetry docs
	[ ] Memcachex - Telemetry docs
	[ ] Nebulex - Telemetry docs
	[ ] Horde - Telemetry docs
	[ ] Cachex - (Need to open up PR)
	[ ] Quantum - Telemetry docs
	[ ] ETS - Erlang docs

Database cron based metrics:
	[ ] Postgres (https://github.com/pawurb/ecto_psql_extras for inspiration)
	[ ] Mnesia (https://github.com/deadtrickster/prometheus.erl/blob/master/src/collectors/mnesia/prometheus_mnesia_collector.erl for inspiration)
	[ ] MySQL (https://github.com/prometheus/mysqld_exporter for inspiration)
	[ ] Redis (https://github.com/oliver006/redis_exporter for inspiration)
	[ ] MongoDB (https://github.com/percona/mongodb_exporter for inspiration)

PromEx.BucketGenerator 
    



      
This module provides functions to generate histogram bucket ranges.
The lists of buckets that can be generated are either linear
or exponential.

PromEx.Config 
    



      
This module defines a struct that contains all of the fields necessary to configure
an instance of PromEx.
While this module does not directly access your Application config, PromEx will call the
PromEx.Config.build/1 function directly with the contents of Application.get_env(:your_otp_app, YourPromEx.Module). As
such, this is an appropriate place to talk about how you go about configuring PromEx via your Application config.
By default, you can run PromEx without any additional configuration and PromEx will fall back on some sane defaults. Specifically,
if you were to not add any configuration to your config.exs, dev.exs, prod.exs, etc files it would be the same as setting the
following config:
config :web_app, WebApp.PromEx,
  disabled: false,
  manual_metrics_start_delay: :no_delay,
  drop_metrics_groups: [],
  grafana: :disabled,
  metrics_server: :disabled
In this configuration, the Grafana dashboards are not uploaded on application start, and a standalone HTTP metrics server is not
started. In addition, the PromEx.ManualMetricsManager is started without any time delay, and all metrics groups from all the plugins
are registered and set up.
If you would like to set up PromEx to communicate with Grafana, your config would look something like:
config :web_app, WebApp.PromEx,
  grafana: [
    host: "http://localhost:3000",
    username: "<YOUR_USERNAME>",  # Authenticate via Basic Auth
    password: "<YOUR_PASSWORD>",
    auth_token: "<YOUR_AUTH_TOKEN_HERE>", # Or authenticate via API Token
    upload_dashboards_on_start: true # This is an optional setting and will default to `true`
  ]
If you would like PromEx to start a standalone HTTP server to serve your aggregated metrics, you can leverage the :metrics_server
config:
config :web_app, WebApp.PromEx,
  metrics_server: [
    port: 4021,
    path: "/metrics", # This is an optional setting and will default to `"/metrics"`
    protocol: :http, # This is an optional setting and will default to `:http`
    pool_size: 5, # This is an optional setting and will default to `5`
    cowboy_opts: [], # This is an optional setting and will default to `[]`
    auth_strategy: :none # This is an optional and will default to `:none`
  ]
If you would like the metrics server to be protected behind some sort of authentication, you can configure your :metrics_server
like so:
config :web_app, WebApp.PromEx,
  metrics_server: [
    port: 4021,
    auth_strategy: :bearer,
    auth_token: "VGhpcyBpcyBzdXBlciBzZWNyZXQuLi5kb24ndCBkZWNvZGUgbWU="
  ]

  
    
  
  Option Details


	:disabled - This option will disable the PromEx supervision tree entirely and will not
start any metrics collectors. This is primarily used for disabling PromEx during testing. Default
value: false

	:manual_metrics_start_delay - Manual metrics are gathered once on start up and then only when
you call PromEx.ManualMetricsManager.refresh_metrics/1. Sometimes, you may have metrics
that require your entire supervision tree to be started in order to fetch accurate data.
This option will allow you to delays the initial metrics capture of the
ManualMetricsManager by a certain number of milliseconds or the :no_delay atom if you
want the metrics to be captured as soon as the ManualMetricsManager starts up. Default
value: :no_delay

	:drop_metrics_groups - A list of all the metrics groups that you are not interested in
tracking. For example, if your application does not leverage Phoenix channels at all but
you still would like to use the PromEx.Plugins.Phoenix plugin, you can pass
[:phoenix_channel_event_metrics] as the value to :drop_metrics_groups and that set of
metrics will not be captured. Default value: []

	ets_flush_interval - This value denotes how often the metrics ETS table is compacted. In order
to keep things performant and as low-overhead as possible, Telemetry metrics are buffered up in
ETS until a request is made to retrieve metrics from the PromEx process. If no requests come in
to extract the metrics, the ETS table can grow infinitely. Luckily, PromEx bundles a GenServer
that periodically compacts ETS. This config value determines how often ETS should be compacted.
Default value: 7_500

	:grafana - This key contains the configuration information for connecting to Grafana. Its
configuration options are:
	:host - The host address of your Grafana instance. In order for PromEx to communicate with
Grafana this value should be in the format protocol://host:port like http://localhost:3000
for example.

	:username - The username that was created in Grafana so that PromEx can upload dashboards
via the API.

	:password - The password that was created in Grafana so that PromEx can upload dashboards
via the API.

	:auth_token - The auth token that was created in Grafana so that PromEx can upload dashboards
via the API.

	:upload_dashboards_on_start - Using the config values that you set in your application config
(config.exs, dev.exs, prod.exs, etc) PromEx will attempt to upload your Dashboards to
Grafana using Grafana's HTTP API.

	:folder_name - The name of the folder that PromEx will put all of the project dashboards in.
PromEx will automatically generate a unique ID for the folder based on the project's otp_app
value so that it can access the correct folder in Grafana. This also makes sure that different
Elixir projects running in the same cluster and publishing dashboards to Grafana do not collide
with one another. If no name is provided, then the dashboards will all be uploaded to the default
Grafana folder.

	:annotate_app_lifecycle - By enabling this setting, PromEx will leverage the Grafana API to annotate
when the application was started, and when it was shut down. By default this is disabled but if you
do enable it, no action is required from you in order to display these events on the dashboards. The
annotations will automatically contain the necessary tags to only display on the PromEx dashboards.
The annotation will include information including:
	Hostname
	OTP app name
	App version
	Git SHA of the last commit (if the GIT_SHA environment variable is present)
	Git author of the last commit (if the GIT_AUTHOR environment variable is present)


	:finch_pools - (optional) A map that will be passed to Finch.start_link/1 as the :pools key,
which can be used to configure protocol, pool size, HTTP host headers, proxy server, etc.



	:grafana_agent - This key contains the configuration information for running GrafanaAgent via a
port in order to push metrics to a Prometheus instance via remote_write functionality:

  
    
  
  Environment dependencies


If your application is running inside of an Alpine Linux container (or any environment that
is based on musl as opposed to
glibc, be sure to add libc6-compat to to your list
of packages. In addition, you'll also need bash running, as this port is wrapped by a
bash script.
For example, in a Dockerfile you would add:
RUN apk add --no-cache bash libc6-compat

	:version - The version of GrafanaAgent that you want to run. This is a string denoting the
GrafanaAgent release version. Below are the supported versions (the downloaded artifacts
are validated against their known SHA256 values so that you can be sure you are not downloading
any malicious binaries and running them). By default, PromEx will use the result of
PromEx.GrafanaAgent.Downloader.default_version() if no value is provided.
	Supported versions are ["0.42.0"]


	:working_directory - In order to leverage the GrafanaAgent functionality, PromEx needs to have
read/write access to a directory in order to download and copy the GrafanaAgent binary. This is the
full path to that directory.

	:config_opts - The configuration file that GrafanaAgent is started with. This option
can either accept an MFA that will return a string of the full path where the YAML configuration
file is, or a keyword list with options so that PromEx can generate a config file for you. If you
take the route where PromEx generates a config file for you, you must provide the following
options:
	:metrics_server_path - The path where the Prometheus metrics are exposed.

	:metrics_server_port - The port that the metrics server is running on.

	:metrics_server_scheme - Whether the app reachable via HTTPS or HTTP (default is https).

	:metrics_server_host - The host to scrape for metrics.

	:instance - This value denotes what instance the metrics are associated with. This value
is a string and defaults to the hostname.

	:job - This value denotes what job the metrics are associated with. This value
is a string and defaults to the otp_app.

	:agent_port - What port should GrafanaAgent run on.

	:grpc_port - What port should GrafanaAgent gRPC server run on.

	:scrape_interval - How often should GrafanaAgent scrape the application. The default is 15s.

	:bearer_token - The bearer token that GrafanaAgent should attach to the request to your app.

	:log_level - The logging level for GrafanaAgent.

	:prometheus_url - The url to your Prometheus instance.

	:prometheus_username - The username to the hosted Prometheus instance

	:prometheus_password - The password to the hosted Prometheus instance

	:template_file - The full path to the template used to render the agent config file.

	all of these keys, and any additional ones, will be provided as EEx vars to the template file.





	:metrics_server - This key contains the configuration information needed to run a standalone
HTTP server powered by Cowboy. This server provides a lightweight solution to serving up PromEx
metrics. Its configuration options are:
	:port - The port that the Cowboy HTTP server should run on.

	:path - The path that the metrics should be accessible at.

	:protocol - The protocol that the metrics should be accessible over (:http or :https).

	:pool_size - How many Cowboy processes should be in the pool to handle metrics related requests.

	:auth_strategy - What authentication strategy should be used to authorize requests to your metrics. The
Supported strategies are :none, :bearer, and :basic. Depending on what strategy is selected, you
will need to also add additional config values. For :none (which is the default), no additional
information needs to be provided. When using a :bearer strategy, you'll need to provide a :auth_token
config value. When using :basic strategy you'll need to provide :auth_user and :auth_password values.

	:auth_token - When using a :bearer authentication strategy, this field is required to validate the
incoming request against a valid auth token.

	:auth_user - When using a :basic authentication strategy, this field is required to validate the
incoming request against a valid user.

	:auth_password - When using a :bearer authentication strategy, this field is required to validate the
incoming request against a valid password.

	:cowboy_opts - A keyword list of any additional options that should be passed to Plug.Cowboy (see
docs for more information https://hexdocs.pm/plug_cowboy/Plug.Cowboy.html). The :port and
:transport_options options are handled by PromEx via the aforementioned config settings and so
adding them again here has no effect.

    PromEx.DashboardRenderer - PromEx v1.10.0
    
    

    


  
  

    
PromEx.DashboardRenderer 
    



      
This module is used to read dashboard definitions, render EEx dashboards,
and ensure that requested files actually exist

    PromEx.DashboardUploader - PromEx v1.10.0
    
    

    


  
  

    
PromEx.DashboardUploader 
    



      
This GenServer is responsible for uploading the configured PromEx module
dashboards to Grafana. This is a transient process and will terminate after
the dashboards have been successfully uploaded. It requires the name of the
PromEx module as an option so that it can look into the application
config for the appropriate Grafana settings. For example, if the name of the
PromEx module is WebApp.PromEx, then your config should provide the following
settings:
config :web_app, WebApp.PromEx,
  grafana_host: "<YOUR HOST ADDRESS>",
  grafana_auth_token: "<YOUR GRAFANA AUTH TOKEN>"

    PromEx.Debug - PromEx v1.10.0
    
    

    


  
  

    
PromEx.Debug 
    



      
This is a convenience module used for debugging and introspecting
telemetry events. Primarily used to ease the development of
PromEx itself.

    PromEx.ETSCronFlusher - PromEx v1.10.0
    
    

    


  
  

    
PromEx.ETSCronFlusher 
    



      
This module is used to regularly flush ETS of any buffered distribution
type metrics (see https://github.com/beam-telemetry/telemetry_metrics_prometheus_core/blob/main/lib/core.ex#L25-L28)
for more information. At the moment the flush interval is not configurable
but that could change in the future.

    PromEx.GrafanaAgent - PromEx v1.10.0
    
    

    


  
  

    
PromEx.GrafanaAgent 
    



      
This GenServer is responsible for starting the Grafana Agent
binary via a port and ensuring that it stays up and running.

    PromEx.GrafanaAgent.ConfigRenderer - PromEx v1.10.0
    
    

    


  
  

    
PromEx.GrafanaAgent.ConfigRenderer 
    



      
This module is used to render the YAML configuration file for
GrafanaAgent.

    PromEx.GrafanaAgent.Downloader - PromEx v1.10.0
    
    

    


  
  

    
PromEx.GrafanaAgent.Downloader 
    



      
This module is responsible for downloading the GrafanaAgent binary.

PromEx.GrafanaClient 
    



      
This module is used by the Mix tasks that are available in PromEx to update
dashboards in Grafana and also by the PromEx.DashboardUpdater to update
dashboards automatically on application initialization.
Dashboard models:
https://grafana.com/docs/grafana/latest/dashboards/json-model/

    PromEx.GrafanaClient.Connection - PromEx v1.10.0
    
    

    


  
  

    
PromEx.GrafanaClient.Connection 
    



      
This struct encapsulates all of the data necessary
to connect to a Grafana instance.

    PromEx.GrafanaClient.DashboardChecker - PromEx v1.10.0
    
    

    


  
  

    
PromEx.GrafanaClient.DashboardChecker 
    



      
This module is used to validate Grafana dashboard to ensure that
they adhere to certain style and structure requirements.

      




  

  
    
    PromEx.LifecycleAnnotator - PromEx v1.10.0
    
    

    


  
  

    
PromEx.LifecycleAnnotator 
    



      
This GenServer is responsible to keeping track of the life cycle
of the application and sending annotation requests to Grafana
when the application starts and when it terminates. It will
include things in the message like:
	Hostname
	OTP app name
	App version
	Git SHA of the last commit (if the GIT_SHA environment variable is present)
	Git author of the last commit (if the GIT_AUTHOR environment variable is present)


      


      
        Summary


  
    Functions
  


    
      
        
  
    
    PromEx.ManualMetricsManager - PromEx v1.10.0
    
    

    


  
  

    
PromEx.ManualMetricsManager 
    



      
This GenServer is responsible to keeping track of all the manual
metrics from your configured plugins. It will fetch metrics initially
when starting (either immediately or after a configured delay). At that
point if you would like to refresh your metrics data points call the
refresh_metrics/1 function.

    PromEx.MetricTypes.Event - PromEx v1.10.0
    
    

    


  
  

    
PromEx.MetricTypes.Event 
    



      
This struct defines the fields necessary to export a group of
standard metrics from a plugin.

    PromEx.MetricTypes.Manual - PromEx v1.10.0
    
    

    


  
  

    
PromEx.MetricTypes.Ma